
#include <WiFi.h>
#include <WebServer.h>
#include <IRremoteESP8266.h>
#include <IRsend.h>
#include <IRutils.h>

// WiFi
const char* ssid     = "DIN_WIFI_SSID";
const char* pass     = "DITT_WIFI_LOSENORD";

// Välj GPIO för IR-sändaren (transistor-basen).
#define IR_LED_PIN 4

IRsend irsend(IR_LED_PIN);
WebServer server(80);

// 12 IR-koder i en array
static const uint32_t IR_CODES[12] = {
  0xFF22DD,        // Kod 1 A

  0xFFE01F,        // Kod 2 B
  0xFF906F,        // Kod 3 C
  0xFF02FD,        // Kod 4 D
  0xFF9867,        // Kod 5 E
  0xFF6897,        // Kod 6 F
  0xFF30CF,        // Kod 7 G
  0xFF18E7,        // Kod 8 H
  0xFF7A85,        // Kod 9 I
  0xFF10EF,        // Kod 10 J
  0xFF38C7,        // Kod 11 U
  0xFF5AA5,        // Kod 12 L
};

// Knappnamn
static const char* KNAPP_TITLES[12] = {
  "-   0 -",

  "-  30 -",
  "-  60 -",
  "-  90 -",
  "- 120 -",
  "- 150 -",
  "- 180 -",
  "- 210 -",
  "- 240 -",
  "- 270 -",



  "- 300 -",
  "- 330 -",
};

// Skicka IR (NEC-protokoll som exempel)
void sendIrCodeNEC(uint32_t code, uint16_t bits = 32) {
  irsend.sendNEC(code, bits);
}

// Huvudsidan med 12 knappar
void handleRoot() {
  String html = R"rawliteral(
<!DOCTYPE html>
<html lang="sv">
<head>
  <meta charset="UTF-8">
  <title>IR Web Controller</title>
  <style>
    body { font-family: sans-serif; }
    .container { max-width: 600px; margin: auto; text-align: center; }
    button {
      display: inline-block;
      margin: 5px;
      padding: 10px 20px;
      font-size: 1em;
      cursor: pointer;
    }
  </style>
</head>
<body>
  <div class="container">
    <h1>IR Web Controller</h1>
    <p>Klicka på en knapp för att skicka en IR-kod.</p>
)rawliteral";

  for (int i = 0; i < 12; i++) {
    html += "<button onclick=\"location.href='/send?index=" + String(i) + 
"'\">";
    html += KNAPP_TITLES[i];
    html += "</button>";
  }

  html += R"rawliteral(
  </div>
</body>
</html>
)rawliteral";

  server.send(200, "text/html", html);



}

// Skicka vald IR-kod och återgå till "/" efter 2 sek
void handleSend() {
  if (!server.hasArg("index")) {
    server.send(400, "text/plain", "Fel: ingen 'index'-parameter");
    return;
  }
  int idx = server.arg("index").toInt();
  if (idx < 0 || idx >= 12) {
    server.send(400, "text/plain", "Fel: ogiltig 'index'");
    return;
  }

  uint32_t codeToSend = IR_CODES[idx];
  sendIrCodeNEC(codeToSend);

  // Bekräftelsesida m. meta refresh -> återgår till "/"
  String response = R"rawliteral(
<!DOCTYPE html>
<html lang="sv">
<head>
  <meta charset="UTF-8">
  <meta http-equiv="refresh" content="2; url=/" />
  <title>IR skickad</title>
</head>
<body>
  <h1>IR-kod skickad!</h1>
  <p>Återgår till huvudsidan om 2 sekunder...</p>
)rawliteral";

  response += "<p>Knapp: " + String(KNAPP_TITLES[idx]) + "</p>";
  response += "<p>Kod (hex): 0x" + String(codeToSend, HEX) + "</p>";

  response += R"rawliteral(
</body>
</html>
)rawliteral";

  server.send(200, "text/html", response);
}

void setup() {
  Serial.begin(115200);
  delay(500);

  // WiFi
  WiFi.mode(WIFI_STA);
  WiFi.begin(ssid, pass);



  while (WiFi.status() != WL_CONNECTED) {
    delay(500);
    Serial.print(".");
  }
  Serial.println("\nWiFi anslutet!");
  Serial.print("IP-adress: ");
  Serial.println(WiFi.localIP());

  // Start IR-sändare
  irsend.begin();

  // Webbserver
  server.on("/", handleRoot);
  server.on("/send", handleSend);
  server.begin();
  Serial.println("HTTP-server startad.");
}

void loop() {
  server.handleClient();
}


